

1. <program> → <prolog> <eol_m> <functions>
2. <prolog> → package main
3. <eol_m> → EOL <eol_r>
4. <eol_r> → EOL <eol_r>
5. <eol_r> → eps
6. <functions> → <func> <function_n>
7. <function_n> → <func> <function_n>
8. <function_n> → eps
9. <func> → func ID (<arguments>) <func_return> <cmd_block>
10. <arguments> → ID <type> <arguments_n>
11. <arguments_n> → , ID <type> <arguments_n>
12. <arguments_n> → eps
13. <type> → int
14. <type> → float64
15. <type> → string
16. <type> → bool
17. <arguments> → eps
18. <func_return> → (<f_type>)
19. <f_type> → <type> <r_type_n>
20. <f_type> → eps
21. <r_type_n> → , <type> <r_type_n>
22. <r_type_n> → eps
23. <func_return> → eps
24. <cmd_block> → { EOL <commands> } EOL
25. <commands> → <cmd> EOL <commands>
26. <commands> → eps
27. <cmd> → ID <statement>
28. <statement> → <id_mul> = <assignment>
29. <id_mul> → , ID <id_mul>
30. <id_mul> → eps
31. <assignment> → <expression> <expr_n>
32. <expr_n> → , <expression> <expr_n>
33. <expr_n> → eps
34. <assignment> → ID (<arguments_fc>)
35. <statement> → (<arguments_fc>)
36. <statement> → := <expression>
37. <statement> → <unary> <expression>
38. <unary> → +=
39. <unary> → -=
40. <unary> → *=
41. <unary> → /=
42. <arguments_fc> → <expression> <expr_n>
43. <arguments_fc> → eps
44. <cmd> → if <expression> <cmd_block> <if_else>
45. <if_else> → else <if_else_st>
46. <if_else> → eps
47. <if_else_st> → <cmd_block>
48. <if_else_st> → if <expression> <cmd_block> <if_else>
49. <cmd> → for <for_definition> ; <expression> ; <for_assignment> <cmd_block>
50. <for_definition> → ID := <expression>
51. <for_definition> → eps
52. <for_assignment> → ID <id_mul> = <assignment>
53. <for_assignment> → eps
54. <cmd> → <return_cmd>
55. <return_cmd> → return <return_stat>
56. <return_stat> → <expression>
57. <return_stat> → eps
58. <return_cmd> → eps